

HARDWARE-SOFTWARE TAPE CONDITIONER FOR TRS-80 COMPUTERS

BY CASS R. LEWART

Reshaping and retiming data pulses ends cassette-tape loading problems for BASIC and SYSTEM data

WHEN difficulties occur in loading data from a cassette tape into a microcomputer, it is usually because the commercial tapes being used are poorly duplicated. This is further compounded by the quality of the tape itself and even the inadequacies of home tape machines. In the case of the popular TRS-80, a narrow tape level setting range and fussy timing requirements exacerbate the problem. Though the "Peak-Reading Meter" in the February 1980 issue of POPULAR ELECTRONICS enables one to set the proper level quickly, it does not correct for poorly shaped pulses or timing jitter, both of which are major obstacles to successful loads. The Tape Regenerator project described here has been designed for this purpose.

The Regenerator is an advanced breed of tape-conditioning device. It is for use with TRS-80 Level II BASIC and machine language (SYSTEM) programs. Unlike other commercially available conditioners, it uses both hardware *and* software. As a result, the computer itself is used for curing timing problems. This permits properly timed backup copies to be made on a second recorder, which, without internal retiming, would produce backup copies that retain or worsen timing jitter.

How It Works. The ideal signal waveform and typical "good" and "poor" waveforms found on commercial copies of Level II programs are shown in Fig. 1. As shown in Fig. 1C, superimposed noise, power-line hum, amplitude distortion, and ringing and displacement of the data pulse relative to the clock pulse (timing jitter) can make it likely that the computer will lose bits. And a single lost bit, of course, makes the entire program useless.

Once a BASIC program has been properly loaded, a back-up copy of it can be made using the CSAVE command. Similarly, a backup copy of a machine language (SYSTEM) program can be



Fig. 1. Waveform (A) is "ideal." That at (B) is a good recording, while those in (C) show various distortions that can occur.

made with a monitor program, e.g. Radio Shack T-BUG. However, the original program must be readable and short enough to fit into memory without overwriting a monitor program.

The Tape Regenerator and the associated program, DUB3 (see Table), overcome these limitations by reshaping and retiming the pulses to produce new tape that the TRS-80 can easily read.

Multiple-segment programs (some programs feature a separate loader) or several programs on a single tape pose no problems for the Regenerator either. Operating on one pair of clock/data pulses at a time without storing the whole program in RAM as monitor programs do, the Regenerator allows even a TRS-80 computer with only 4K of RAM to make back-up copies of arbitrarily large programs and handle tapes containing multiple programs in a single cassette pass.

To test soundness of the Tape Regenerator design, the author created five generations of the same program and found that the fifth-generation tape would load as easily as the original.

Circuit Description. The circuit shown in Fig. 2 reshapes the clock and data pulses received from the tape recorder and feeds them to the computer. Audio transformer *T1* provides dc isolation between the tape recorder and the Regenerator circuit. Switch *S1* and diodes *D1* and *D2* allow selection of the "better" half of the pulse (see Control Adjustments later on), while zener diode *D3* and transistor *Q1* further shape the incoming signal. One OR gate in *IC1* decodes the IN command from the computer (via the *P1* connector) indicating that the computer is ready to accept data. When the command occurs, *IC1* (pin 8) then activates *IC2* (via pin 15) to allow the amplified tape recorder signals to pass via the connector *P1* and the expansion port to the computer data bus. Indicator *LED2* and optional meter *M1* indicate that the tape recorder is sending data at the proper level, and *LED3* glows when regenerating program DUB3 is up and running. If no back-up copies are required, *J2* provides a "quick and dirty" direct output to the TRS-80 via the tape-recorder plug. The signal at this point is not retimed and is only partially reshaped. This limited processing may make a tape readable.

Power for the circuit (Fig. 3), is provided by transformer *T2* in conjunction with voltage regulator *IC3*, bridge circuit *RECT1*, and capacitors *C1* and *C2*. Power on is indicated by *LED1*.



Fig. 2. The Tape Regenerator is controlled by signals from the TRS-80 that are determined by the DUB3 program. A second recorder is connected to the auxiliary output at jack J2.

PARTS LIST

C1—200-μF, 35-V electrolytic
C2,C3—0.1-μF disc capacitor
D1,D2—1N914 silicon diode
D3—3.9-V zener diode
IC1—74LS32 quad OR gate
IC2—74LS368 hex tri-state buffer
IC3—7805 +5-volt regulator
J1,J2—Miniature phone jack
LED1,LED2,LED3—Red light-emitting diode
M1—1-mA meter (Radio Shack 270-1752 or similar)
P1—2 × 20 edge connector on 0.1" centers to fit expansion port on TRS-80 keyboard or expansion interface
Q1—Npn Darlington Transistor (HEPS9100 or similar)
R1,R3,R5—270-Ω, ½-W resistor
R2—10-kΩ, ½-W resistor
R4,R6—820-Ω, ½-W resistor (optional)
RECT1—50-V, 1-A bridge rectifier
S1,S2—Spdt switch
T1—Audio transformer (Radio Shack 273-1380 or similar)
T2—12-volt transformer (Radio Shack 273-1385 or similar)
Misc.—Suitable enclosure, 2 DIP sockets (14 pin and 16 pin) line cord, solder, etc.

Note—The following is available from Southwest Technical Products, 219 W. Rhapsody, San Antonio, TX 78216, Dept. TR-1: complete kit of parts including a pc board and listing of DUB3 program in BASIC to allow POKEing into memory, but excluding M1, at $29.95 postpaid. Texas residents, add 4% sales tax.



Fig. 3. The power supply is a simple rectifier-regulator circuit as shown here.

Construction. The circuit can be assembled on a small pc board, or Wire-Wrap techniques can be used. Keep all leads as short as possible. As only four contacts are used on the 40-pin TRS-80 connector, the remaining pins can be expanded with a screwdriver to make plug insertion easier. Before turning the power on, recheck all wiring, especially the leads to the computer expansion port.

Control Adjustments. To properly read a poorly recorded tape, polarity switch *S1* of the Tape Regenerator and the volume control of the tape recorder have to be set. These settings will vary from tape to tape. However, back-up tapes made by the Tape Regenerator or CSAVE command should all work with the same settings.

To determine the proper settings for an unknown tape, observe *LED2* and milliameter *M1*, or connect a scope to *TP1*. Play the tape at medium setting of the recorder volume control. Flip polarity switch *S1* and leave it in the position corresponding to a stronger signal as evidenced by a brighter LED, higher reading on the meter, or a cleaner pulse display on the scope. Optimum setting of the recorder playback level is found by advancing the volume control until *LED2* glows brightly, then backing off slightly. If optional meter *M1* is included in the circuit (see Fig. 2) adjust the volume control for a reading between 0.5 and 0.6 mA. As a final alternative, connect an oscilloscope to *TP1* and adjust the control for the cleanest, widest pulses you can.

Regenerator Program DUB3. The reshaping and retiming of Level-II clock and data pulses as received via the *P1* connector is performed by the DUB3 program.

After clearing the screen and displaying a message, the DUB3 program searches for a clock pulse. When one is found, the time interval to a second pulse is checked to make sure that the first was not a spurious transient. When the clock pulse is confirmed, it is output after a 200-µs delay using subroutine OUTPUT. This subroutine produces a clean signal lasting 300 µs as shown in Fig. 1A. A search for the data pulse now begins. The delay of 500 µs excludes any residual ringing from the preceding clock pulse. If no data pulse is found during the following 700-µs window, a search for the next clock pulse begins. If a data pulse is detected in the window, it is checked again to exclude a transient



Fig. 4. Timing diagram of DUB3 program. Input at TP1 is at top. Below is output from computer.



Fig. 5. Electrical interconnection of the Tape Regenerator to the TRS-80 computer and two tape recorders.

TABLE—LISTING OF DUB3 PROGRAM

```
00100  PORT    EQU   00H
00110  CENTER  EQU   04H
00120  HIGH    EQU   05H
00130  LOW     EQU   06H
00140  DL100   EQU   0CH
00150  DL150   EQU   12H
00160  LEN1    EQU   14H
00170  DL200   EQU   1AH
00180  DL700   EQU   22H
00190  LEN2    EQU   22H
00200  CASS    EQU   0FFH
00210  CLS     EQU   01C9H
00220  LINE1   EQU   3DD6H
00230  LINE2   EQU   3E0FH
00240          ORG   4A00H
00250  DUB3    CALL  CLS        ;CLEAR SCREEN
00260          LD    HL,TAB1    ;DISPLAY LINE1
00270          LD    DE,LINE1
00280          LD    BC,LEN1
00290          LDIR
00300          LD    HL,TAB2    ;DISPLAY LINE2
00310          LD    DE,LINE2
00320          LD    BC,LEN2
00330          LDIR
00340  START   LD    B,DL100    ;100 MICS DELAY
00350  DEL0    DJNZ  DEL0
00360  SRCHC   IN    A,(PORT)   ;START CLOCK PULSE SEARCH
00370          RRA
00380          JR    NC,SRCHC   ;CLOCK PULSE FOUND?
00390          IN    A,(PORT)   ;YES, TRANSIENT ONLY?
00400          RRA
00410          JR    NC,SRCHC   ;YES, KEEP SEARCHING
00420          LD    B,DL200    ;NO, SET 200 MICS DELAY
00430  DEL1    DJNZ  DEL1
00440          CALL  OUTPUT     ;PUT OUT CLOCK PULSE
00450          LD    B,DL700    ;START 700 MICS READ WINDOW
00460  SRCHD   IN    A,(PORT)   ;SEARCH FOR DATA PULSE
00470          RRA
00480          JR    C,FOUND1   ;FOUND?
00490          DJNZ  SRCHD      ;NO, WINDOW TIMED OUT?
00500          JR    START      ;YES, SEARCH FOR CLOCK PULSE
00510  FOUND1  IN    A,(PORT)   ;TRANSIENT ONLY?
00520          RRA
00530          JR    C,FOUND2   ;NO
00540          DJNZ  SRCHD      ;YES, WINDOW TIMED OUT?
00550          JR    START      ;YES SEARCH FOR CLOCK PULSE
00560                           ;DATA PULSE FOUND,
00570  FOUND2  INC   IX         ;WASTE 10 CYCLES
00580          BIT   3,(HL)     ;WASTE 12 CYCLES
00590          DJNZ  FOUND2     ;WINDOW TIMED OUT?
00600          CALL  OUTPUT     ;YES, PUT OUT DATA PULSE
00610          JR    START      ;SEARCH FOR CLOCK PULSE
00620  OUTPUT  LD    A,HIGH     ;PULSE OUTPUT
00630          OUT   (CASS),A   ;PULSE HIGH
00640          LD    B,DL150    ;150 MICS DELAY
00650  DEL2    DJNZ  DEL2
00660          LD    A,LOW
00670          OUT   (CASS),A   ;PULSE LOW
00680          LD    B,DL150    ;150 MICS DELAY
00690  DEL3    DJNZ  DEL3
00700          LD    A,CENTER
00710          OUT   (CASS),A   ;RESTORE TO CENTER
00720          RET
00730  TAB1    DEFM  'TAPE BACK-UP PROGRAM'
00740  TAB2    DEFM  'COPYRIGHT (C) 1980 CASS R. LEWART'
00750          END   DUB3
```

and, if confirmed, is output at the end of the 1-ms interval that started at the beginning of the preceding clock pulse (Fig. 4).

A data pulse appearing any time between 500 μs and 1.2 ms after a clock pulse is thus correctly retimed to occur exactly 1 ms after the clock pulse. After a 100-μs delay, the program continues with the search for the next clock pulse. The DUB3 program can be loaded using the Radio Shack Editor/Assembler or by keying in the Z80 instructions. For a BASIC version of the DUB3 program which will POKE the instructions into memory see the Parts List.

Operating Instructions. The electrical interconnection between the Tape Regenerator, both tape recorders and the computer is shown in Fig. 5. Turn the computer and Tape Regenerator power off when plugging or unplugging the 40-pin connector at the rear of the TRS-80 keyboard.

If you have the Expansion Interface connected to your computer, use the Expansion Port on the left side of the Expansion Interface instead of the Expansion Port at the rear of the keyboard. When power is applied to computer and Tape Regenerator, *LED1* should glow and the MEMORY SIZE? prompt should appear on the video monitor. If the prompt does not appear, check connections, in particular the 40-pin connector, between the Tape Regenerator and TRS-80. Load the DUB3 program and run it. Light *LED3* (PGM ON) should glow as long as DUB3 is running. The program is in an infinite loop and will run until you depress the RESET pushbutton on the rear of the TRS-80, or turn the computer off.

For initial adjustment, start reading tape from tape recorder 1 and set polarity switch *S1* and the Tape Recorder volume control as explained under "Control Adjustments."

Rewind tape recorder #1 and start it in the play mode while starting tape recorder #2 in the record mode with a clean tape. When the program on tape recorder #1 is finished, *LED2* (VOL) will extinguish and meter *M1* will indicate close to zero. This is the signal for you to turn both tape recorders off. You can continue with as many tapes as desired. When finished, open *S2* to turn the Tape Regenerator power off, press the RESET button to return to BASIC, or turn the computer off. The 40-pin connector *P1* can be left plugged permanently into the Expansion Port as it does not interfere with the normal computer operation. If the pulse amplitude on the original tape is very unsteady or the pulses are imbedded in noise, regenerating the tape may not be possible. ◇

BUILDING THIS CIRCUIT IS NOT RECOMMENDED

THE TECHNOLOGY IS TOO OLD - MUCH BETTER IS NOW AVAILABLE

THIS CIRCUIT IS PRESENTED ONLY BECAUSE STUDY OF IT CAN GIVE A PERSON BETTER UNDERSTANDING OF THE THEORY BEHIND MODERN TECHNOLOGY